Ratio Mathematica 10
(1996), 9 - 22

ELEMENTI DI GEOMETRIE FINITE

Stefania Ferri*

SUNTO - Si fa vedere come si può costruire la geometria analitica su un piano affine finito. Vengono riportati risultati classici che però sono stati esposti in modo da essere compresi dagli alunni delle Scuole Medie Secondarie.

INTRODUZIONE

La geometria analitica che si studia nelle Scuole Medie Superiori è la rappresentazione di punti, rette e curve nel piano affine reale. Cioè le coordinate dei punti, i coefficienti delle equazioni delle rette e delle curve sono numeri reali. Talvolta tale piano viene complessificato nel senso che si considerano anche i punti a coordinate numeri complessi (quando per esempio si vogliono trovare le intersezioni di una circonferenza con una retta esterna).

Si può però, come è noto, considerare il caso in cui tali coordinate e tali coefficienti appartengano, invece che al campo dei numeri reali, ad un campo finito, cioè costituito da un numero finito di elementi. In questa situazione, contrariamente al caso dei reali, i punti del piano sono in numero finito, così le rette e le curve, come anche finito è il numero dei punti di una retta o di una curva.

Si ha così una geometria finita che trova numerosissime applicazioni anche perchè permette, per la risoluzione di molti problemi, l'uso del computer.

Ad esempio ci si può riferire al campo dei resti modulo p, che si indica con Z_p, e cioè al campo i cui elementi sono soltanto 0,1,2,...,p-1 ed in cui p=0, p+1=1, p+2=2,... Si dimostra che in questo caso, se p è primo, o la potenza di un numero primo $q=p^h$, esiste il piano affine in cui le coordinate dei punti e i coefficienti delle curve siano elementi di Z_p. Tale piano si indica con AG(2,q).

---

* Lavoro fatto presso la sezione di Matematica del Dipartimento di Meccanica della Facoltà di Ingegneria di Roma III.
Il presente articolo è stato presentato, in un seminario, a docenti di Scuola Media Superiore.

Si dimostra che in AG(2,q):
- *il numero dei punti q^2;*
- *il numero delle rette è q^2+q;*
- *il numero dei punti di una retta è q;*
- *per ogni punto passano q+1 rette.*

Inoltre per le coniche irriducibili di AG(2,q) si prova che:
- *un'ellisse ha q+1 punti in AG(2,q);*
- *una parabola ha q punti in AG(2,q) (in quanto uno è all'infinito);*
- *un'iperbole ha q-1 punti in AG(2,q) (in quanto due sono all'infinito);*

Le cose vanno in modo diverso a seconda che q sia *pari* o *dispari*, come faremo vedere trattando qualche caso particolare.

Considereremo i casi più semplici e precisamente i casi q=2 e q=3.

IL PIANO AFFINE AG(2,2)

Il piano affine costruito su Z_2 (i cui elementi sono solo 0 ed 1 ed in cui 1=-1 e 2=0), cioè AG(2,2) è tale che:

- *in numero dei punti è $q^2=4$;*
- *il numero dellle rette è $q^2+q=6$;*
- *il numero dei punti di una retta è q=2 punti;*
- *per ogni punti passano q+1=3 rette.*

I punti sono (0,0), (0,1), (1,0) e (1,1) e le rette x=0, y=0, x+1=0, y+1=0, x+y=0 e x+y+1=0.



Si noti che le rette sono costituite da soli due punti. Precisamente

x=0	è costituita da (0,0) e (0,1);
y=0	è costituita da (0,0) e (1,0);
x+1=0	è costituita da (1,0) e (1,1);
y+1=0	è costituita da (0,1) e (1,1);
x+y=0	è costituita da (0,0) e (1,1);
x+y+1=0	è costituita da (1,0) e (0,1).

Si verifica subito che per ogni punto passano tre rette (per (0,0), ad esempio, passano le tre rette x=0, y=0, x+y=0 e così via).

Vediamo come si comportano le coniche in AG(2,2). Le cose si discostano dal caso q dispari e valgono anche in AG(2,q) con $q=2^h$.

L'equazione di una conica in AG(2,2) si scrive:

$$a_{11}x^2+a_{22}y^2+a_{12}xy+a_{13}x+a_{23}y+a_{33}=0 \text{ dove } a_{ik}\in Z_2.$$

Se $a_{12}=a_{13}=a_{23}=0$ si ha:

$$a_{11}x^2+a_{22}y^2+a_{33}=0$$

e cioè, poichè in Z_2 ogni elemento è un quadrato, ne segue:

$$(\sqrt{a_{11}}\,x+\sqrt{a_{22}}\,y+\sqrt{a_{33}}\,)^2=0$$

cioè ogni conica è doppiamente degenere cioè è costituita da due rette coincidenti.

Si dimostra facilmente che vale anche il viceversa cioè *se la conica è doppiamente degenere si ha che* $a_{12}=a_{13}=a_{23}=0$.

Una conica è *semplicemente degenere* se è costituita da due rette distinte cioè se il polinomio di secondo grado che la rappresenta è il prodotto di due polinomi di primo grado distinti a coefficienti in Z_2 (o in un ampliamento quadratico di esso).

In AG(2,2) i 6 coefficienti dell'equazione di una conica possono prendere solo i valori 0 ed 1 e quindi il numero di tali coniche sarà dato dalle disposizioni con ripetizione di due elementi a sei a sei. Da tale numero bisogna togliere la sestupla costituita da tutti zeri e quella in cui sono zero i primi cinque coefficienti e le sei che risultano di primo grado, quindi il numero sarà $2^6-8 = 56$.

Si possono avere mediante il seguente programma le equazioni di tutte le coniche di AG(2,2).

```
{QUESTO PROGRAMMA SCRIVE LE EQUAZIONI DELLE}
{CONICHE}

PROGRAM coniche;
USES crt;
VAR i:INTEGER;
FUNCTION eq(index:BYTE):STRING;
VAR
 eqtmp : STRING;
 i    : INTEGER;
BEGIN { eq }
 eqtmp:='';
 FOR i:=1 TO 6 DO
  BEGIN
   IF (index AND 1)=1 THEN
    CASE i OF
     1: eqtmp:='1'+eqtmp;
```

```
        2: BEGIN
            IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
            eqtmp:='y'+eqtmp
          END;
        3: BEGIN
            IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
            eqtmp:='x'+eqtmp
          END;
        4: BEGIN
            IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
            eqtmp:='xy'+eqtmp
          END;
        5: BEGIN
            IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
            eqtmp:='yy'+eqtmp
          END;
        6: BEGIN
            IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
      eqtmp:='xx'+eqtmp
          END
      END;
     index:=index SHR 1
   END;
  IF (LENGTH(eqtmp)>0) THEN eqtmp:=eqtmp+'=0';
  eq:=eqtmp;
END;  { eq }
BEGIN { main }
 FOR i:=8 TO 63 DO
   WRITE(eq(i):20);
 READLN;
END.  { main }
```

Come equazioni si avranno in output:

1) $xy=0$ 2) $xy+1=0$
3) $xy+y=0$ 4) $xy+y+1=0$
5) $xy+x=0$ 6) $xy+x+1=0$
7) $xy+x+y=0$ 8) $xy+x+y+1=0$
9) $y^2=0$ 10) $y^2+1=0$
11) $y^2+y=0$ 12) $y^2+y+1=0$
13) $y^2+x=0$ 14) $y^2+x+1=0$
15) $y^2+x+y=0$ 16) $y^2+x+y+1=0$
17) $y^2+xy=0$ 18) $y^2+xy+1=0$
19) $y^2+xy+y=0$ 20) $y^2+xy+y+1=0$
21) $y^2+xy+x=0$ 22) $y^2+xy+x+1=0$

23) $y^2+xy+x+y=0$	24) $y^2+xy+x+y+1=0$
25) $x^2=0$	26) $x^2+1=0$
27) $x^2+y=0$	28) $x^2+y+1=0$
29) $x^2+x=0$	30) $x^2+x+1=0$
31) $x^2+x+y=0$	32) $x^2+x+y+1=0$
33) $x^2+xy=0$	34) $x^2+xy+1=0$
35) $x^2+xy+y=0$	36) $x^2+xy+y+1=0$
37) $x^2+xy+x=0$	38) $x^2+xy+x+1=0$
39) $x^2+xy+x+y=0$	40) $x^2+xy+x+y+1=0$
41) $x^2+y^2=0$	42) $x^2+y^2+1=0$
43) $x^2+y^2+y=0$	44) $x^2+y^2+y+1=0$
45) $x^2+y^2+x=0$	46) $x^2+y^2+x+1=0$
47) $x^2+y^2+x+y=0$	48) $x^2+y^2+x+y+1=0$
49) $x^2+y^2+xy=0$	50) $x^2+y^2+xy+1=0$
51) $x^2+y^2+xy+y=0$	52) $x^2+y^2+xy+y+1=0$
53) $x^2+y^2+xy+x=0$	54) $x^2+y^2+xy+x+1=0$
55) $x^2+y^2+xy+x+y=0$	56) $x^2+y^2+xy+x+y+1=0$

Per vedere quali di tali coniche sono ellissi, parabole, iperboli non si può seguire il criterio che si applica nel caso del piano affine sui reali in cui l'equazione di una conica si scrive

$a_{11}x^2+a_{22}y^2+2a_{12}xy+2a_{13}x+2a_{23}y+a_{33}=0$ dove $a_{ik}\in\mathbf{R}$

ed in cui si vede se $\Delta=a_{12}^2-a_{11}a_{12}>=<0$ (dove Δ è il discriminante dell'equazione che si ottiene ponendo uguali a zero i termini di 2° grado della suddetta equazione), poichè in Z_2 non si può applicare la formula risolutiva dell'equazione di 2° grado, essendo 2=0.

Basta però ricordare (cfr. Introduzione) che: in AG(2,2) un'ellisse ha tre punti, una parabola ne ha due e un'iperbole ne ha uno.

Ad esempio la conica $x^2+y^2+xy+1=0$ contiene i punti (1,0), (0,1), (1,1) e quindi è un ellisse; la conica $x+y^2+1=0$ contiene i punti (0,1) e (1,0) e quindi è una parabola; la conica $xy+x+y=0$ contiene il punto (0,0) e quindi è un'iperbole.

Si noti ancora come esempio che la conica $x^2+y^2+1=0$ risulta doppiamente degenere $(x+y+1)^2=0$ e la conica $x^2+xy=0$ risulta semplicemente degenere in quanto si spezza nelle rette $x=0$ e $x+y=0$.

Si può scrivere il seguente programma che riconosce se una conica è degenere (cioè si spezza nel prodotto di due delle sei rette di AG(2,2), distinte o coincidenti) e nel caso in cui sia non degenere dice se è un'ellisse, una parabola, un'iperbole.

```
{QUESTO PROGRAMMA RICONOSCE SE UNA CONICA E'}
{DEGENERE E, IN TAL CASO, TROVA LE EQUAZIONI DELLE}
{RETTE IN CUI SI SPEZZA E, NEL CASO IN CUI NON E'}
{DEGENERE, DICE SE E' UN'ELLISSE, UNA PARABOLA OPPURE}
{UN'IPERBOLE.}

PROGRAM riconoscitore;
USES crt;
FUNCTION eq(index:BYTE):STRING;
VAR
 eqtmp : STRING;
 i     : INTEGER;
BEGIN { eq }
 eqtmp:='';
 FOR i:=1 TO 6 DO
  BEGIN
   IF (index AND 1)=1 THEN
    CASE i OF
     1: eqtmp:='1'+eqtmp;
     2: BEGIN
         IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
         eqtmp:='y'+eqtmp
        END;
     3: BEGIN
         IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
         eqtmp:='x'+eqtmp
        END;
     4: BEGIN
         IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
         eqtmp:='xy'+eqtmp
        END;
     5: BEGIN
         IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
         eqtmp:='yy'+eqtmp
        END;
     6: BEGIN
    IF (LENGTH(eqtmp)>0) THEN eqtmp:='+'+eqtmp;
         eqtmp:='xx'+eqtmp
        END
    END;
   index:=index SHR 1
  END;
IF (LENGTH(eqtmp)>0) THEN eqtmp:=eqtmp+'=0';
 eq:=eqtmp;
```

```
END; { eq }
FUNCTION eqprod(eq1,eq2:BYTE):BYTE;
VAR
 m,n,o,p,q,r : BYTE;
BEGIN { eqprod }
 m:=((eq1 SHR 2) AND (eq2 SHR 2)) AND 1;
 n:=((eq1 SHR 1) AND (eq2 SHR 1)) AND 1;
 r:=(eq1 AND eq2) AND 1;
 o:=((eq1 SHR 2) AND (eq2 SHR 1) AND 1)+((eq1 SHR 1) AND (eq2 SHR
2) AND 1);
 p:=((eq1 SHR 2) AND eq2 AND 1)+(eq1 AND (eq2 SHR 2) AND 1);
 q:=((eq1 SHR 1) AND eq2 AND 1)+(eq1 AND (eq2 SHR 1) AND 1);
 o:=o MOD 2;
 p:=p MOD 2;
 q:=q MOD 2;
 eqprod:=(r AND 1) OR ((q SHL 1) AND 2) OR
              ((p SHL 2) AND 4) OR
              ((o SHL 3) AND 8) OR
              ((n SHL 4) AND 16) OR
              ((m SHL 5) AND 32)
END; { eqprod }
FUNCTION evaluate(index,x,y:BYTE):BOOLEAN;
VAR
 v : BYTE;
BEGIN { evaluate }
 IF (x<>0) THEN x:=1;
 IF (y<>0) THEN y:=1;
 v:= ((((index SHR 5) AND 1)*x) +
    (((index SHR 4) AND 1)*y) +
    (((index SHR 3) AND 1)*x*y) +
    (((index SHR 2) AND 1)*x) +
    (((index SHR 1) AND 1)*y) + (index AND 1)) MOD 2;
 IF (v=0) THEN evaluate:=TRUE
       ELSE evaluate:=FALSE
END; { evaluate }
FUNCTION recognize(index:BYTE):STRING;
VAR
 i,j,num : BYTE;
BEGIN { recognize }
 recognize:='';
 num:=0;
 FOR i:=0 TO 1 DO
  FOR j:=0 TO 1 DO
   IF evaluate(index,i,j) THEN num:=num+1;
```

```
 CASE num OF
   1: recognize:='Iperbole';
   2: recognize:='Parabola';
   3: recognize:='Ellisse'
 END
END;  { recognize }
PROCEDURE exclude;
CONST
 rette : ARRAY [1..6] OF BYTE = (4,2,5,3,6,7);
VAR
 i,j,k : BYTE;
 flag  : BOOLEAN;
 f     : TEXT;
BEGIN { exclude }
 ASSIGN(f,'r.asc');
 REWRITE(f);
 FOR i:=8 TO 63 DO
   BEGIN
     WRITE(f,eq(i):20);
     flag:=TRUE;
     FOR j:=1 TO 6 DO
       FOR k:=j TO 6 DO
         IF (eqprod(rette[j],rette[k])=i) THEN
           BEGIN
             flag:=FALSE;
             WRITELN(f,'        [',eq(rette[j]),'] * [',eq(rette[k]),']')
           END;
     IF flag THEN WRITELN(f,'        ',recognize(i))
   END;
 CLOSE(f)
END;  { exclude }
BEGIN { main }
exclude
END.  { main }
```

In output si hanno i seguenti risultati:

1) xy=0	[x=0]*[y=0]
2) xy+1=0	Iperbole
3) xy+y=0	[y=0]*[x+1=0]
4) xy+y+1=0	Iperbole
5) xy+x=0	[x=0]*[y+1=0]
6) xy+x+1=0	Iperbole
7) xy+x+y=0	Iperbole
8) xy+x+y+1=0	[x+1=0]*[y+1=0]

9) $y^2=0$	[y=0]*[y=0]
10) $y^2+1=0$	[y+1=0]*[y+1=0]
11) $y^2+y=0$	[y=0]*[y+1=0]
12) $y^2+y+1=0$	
13) $y^2+x=0$	Parabola
14) $y^2+x+1=0$	Parabola
15) $y^2+x+y=0$	Parabola
16) $y^2+x+y+1=0$	Parabola
17) $y^2+xy=0$	[y=0]*[x+y=0]
18) $y^2+xy+1=0$	Iperbole
19) $y^2+xy+y=0$	[y=0]*[x+y+1=0]
20) $y^2+xy+y+1=0$	Iperbole
21) $y^2+xy+x=0$	Iperbole
22) $y^2+xy+x+1=0$	[y+1=0]*[x+y+1=0]
23) $y^2+xy+x+y=0$	[y+1=0]*[x+y=0]
24) $y^2+xy+x+y+1=0$	Iperbole
25) $x^2=0$	[x=0]*[x=0]
26) $x^2+1=0$	[x+1=0]*[x+1=0]
27) $x^2+y=0$	Parabola
28) $x^2+y+1=0$	Parabola
29) $x^2+x=0$	[x=0]*[x+1=0]
30) $x^2+x+1=0$	
31) $x^2+x+y=0$	Parabola
32) $x^2+x+y+1=0$	Parabola
33) $x^2+xy=0$	[x=0]*[x+y=0]
34) $x^2+xy+1=0$	Iperbole
35) $x^2+xy+y=0$	Iperbole
36) $x^2+xy+y+1=0$	[x+1=0]*[x+y+1=0]
37) $x^2+xy+x=0$	[x=0]*[x+y+1=0]
38) $x^2+xy+x+1=0$	Iperbole
39) $x^2+xy+x+y=0$	[x+1=0]*[x+y=0]
40) $x^2+xy+x+y+1=0$	Iperbole
41) $x^2+y^2=0$	[x+y=0]*[x+y=0]
42) $x^2+y^2+1=0$	[x+y+1=0]*[x+y+1=0]
43) $x^2+y^2+y=0$	Parabola
44) $x^2+y^2+y+1=0$	Parabola
45) $x^2+y^2+x=0$	Parabola
46) $x^2+y^2+x+1=0$	Parabola
47) $x^2+y^2+x+y=0$	[x+y=0]*[x+y+1=0]
48) $x^2+y^2+x+y+1=0$	
49) $x^2+y^2+xy=0$	Iperbole
50) $x^2+y^2+xy+1=0$	Ellisse
51) $x^2+y^2+xy+y=0$	Ellisse
52) $x^2+y^2+xy+y+1=0$	Iperbole

53) $x^2+y^2+xy+x=0$ Ellisse
54) $x^2+y^2+xy+x+1=0$ Iperbole
55) $x^2+y^2+xy+x+y=0$ Ellisse
56) $x^2+y^2+xy+x+y+1=0$ Iperbole

Si noti che il programma non ha specificato tre coniche, perchè esse sono degeneri, ma si spezzano in rette che non appartengono al campo Z_2, in quanto i polinomi che le rappresentano sono irriducibili in Z_2, ma riducibili in un ampliamento algebrico (come succede per i polinomi che sono irriducibili nel campo dei reali, ma riducibili nel campo dei complessi).

IL PIANO AFFINE AG(2,3)

Consideriamo ora AG(2,3) cioè il piano affine costruito su Z_3 e cioè sul campo dei resti modulo 3, i cui elementi sono quindi solo 0, 1, 2 (ed in cui 3=0, 1=-2).

In AG(3,q):

- *i punti sono in numero di $q^2=3^2=9$;*
- *le rette sono in un numero di $q^2+q=3^2+3=12$;*
- *il numero di punti di una retta sono $q=3$;*
- *per un punto passano quattro rette.*

I punti sono:

(0,0), (1,0), (2,0), (0,1), (0,2), (1,1), (2,2), (1,2), (2,1).



Le 12 rette (ognuna delle quali contiene 3 punti) si ottengono dall'equazione $ax+by+c=0$ prendendo i coefficienti appartenenti a Z_3 e togliendo una retta se ha un coefficiente uguale a 2 e si può ottenere da un'altra quando si moltiplicano i coefficienti per 2 e si prendono modulo 3.
Ad esempio $x+2y=0$ si può scrivere $2x+4y=0$ cioè $2x+y=0$ ed essendo tali rette coincidenti, bisogna eliminarne una.

Possiamo allora dire che le 12 rette hanno per equazioni le seguenti:

$x=0$ che è costituita da (0,0), (0,1) e (0,2);
$y=0$ che è costituita da (0,0), (1,0) e (2,0);

x+y=0 che è costituita da (0,0), (1,2) e (2,1);
2x+y=0 che è costituita da (0,0), (1,1) e (2,2);
x+y+1=0 che è costituita da (1,1), (2,0) e (0,2);
2x+2y+1=0 che è costituita da (1,0), (0,1) e (2,2);
2x+1=0 che è costituita da (1,0), (1,1) e (1,2);
x+1=0 che è costituita da (2,0), (2,2) e (2,1);
y+1=0 che è costituita da (0,2), (2,2) e (1,2);
2x+y+1=0 che è costituita da (0,2), (1,0) e (2,1);
2x+y+2=0 che è costituita da (2,0), (0,1) e (1,2);
2y+1=0 che è costituita da (0,1), (1,1) e (2,1).

Per ogni punto passano 4 rette, come subito si verifica (ad esempio per 0 passano le rette x=0, y=0, x+y=0, 2x+y=0).

Si possono scrivere le equazioni suddette tramite il seguente programma.

```
{QUESTO PROGRAMMA SCRIVE LE EQUAZIONI DELLE RETTE}

PROGRAM rette;
USES crt;
VAR
 eq : ARRAY [1..27] OF BOOLEAN;
PROCEDURE elimina1;
VAR
 a,b,c,i : BYTE;
BEGIN { elimina1 }
 i:=1;
 FOR a:=0 TO 2 DO
  FOR b:=0 TO 2 DO
   FOR c:=0 TO 2 DO
    BEGIN
     eq[i]:=TRUE;
     IF ((a=0) AND (b=0)) THEN eq[i]:=FALSE;
     IF ((a=2) AND (b=2) AND (c=2)) THEN eq[i]:=FALSE;
     IF (a=0) THEN
      IF ((b=2) AND (c=2)) THEN eq[i]:=FALSE;
     IF (b=0) THEN
      IF ((a=2) AND (c=2)) THEN eq[i]:=FALSE;
     IF (c=0) THEN
      IF ((a=2) AND (b=2)) THEN eq[i]:=FALSE;
     IF (a=2) THEN
      IF ((b=0) AND (c=0)) THEN eq[i]:=FALSE;
     IF (b=2) THEN
      IF ((a=0) AND (c=0)) THEN eq[i]:=FALSE;
     IF (c=2) THEN
      IF ((a=0) AND (b=0)) THEN eq[i]:=FALSE;
```

```
      i:=i+1
     END
END;  { elimina1 }
PROCEDURE elimina2;
VAR
 a,b,c,aa,bb,cc,i : BYTE;
BEGIN { elimina2 }
 i:=1;
 FOR a:=0 TO 2 DO
  FOR b:=0 TO 2 DO
   FOR c:=0 TO 2 DO
    BEGIN
     IF eq[i] THEN
      IF ((a=2) OR (b=2) OR (c=2)) THEN
       BEGIN
        aa:=(a*2) MOD 3;
        bb:=(b*2) MOD 3;
        cc:=(c*2) MOD 3;
        IF eq[(aa*9)+(bb*3)+cc+1] THEN eq[i]:=FALSE
       END;
     i:=i+1
    END
END;  { elimina2 }
PROCEDURE stampa;
VAR
 a,b,c,i : BYTE;
 f      : TEXT;
BEGIN { stampa }
 ASSIGN(f,'r1.asc');
 REWRITE(f);
 i:=1;
 FOR a:=0 TO 2 DO
  FOR b:=0 TO 2 DO
   FOR c:=0 TO 2 DO
    BEGIN
     IF eq[i] THEN
      BEGIN
       IF (a<>0) THEN
        IF (a=1) THEN WRITE(f,'x')
             ELSE WRITE(f,'2x');
       IF (b<>0) THEN
    BEGIN
         IF (a<>0) THEN WRITE(f,'+');
         IF (b=1) THEN WRITE(f,'y')
```

```
                ELSE WRITE(f,'2y');
            END;
          IF (c<>0) THEN WRITE(f,'+',c);
          WRITELN(f,'=0')
        END;
      i:=i+1
    END;
  CLOSE(f)
END; { stampa }
BEGIN { main }
  elimina1;
  elimina2;
  stampa
END. { main }
```

Per quanto riguarda le coniche esse si possono scrivere come nel caso del piano affine su $\mathbb{R}$:

$$a_{11}x^2+a_{22}y^2+2a_{12}xy+2a_{13}x+2a_{23}y+a_{33}=0$$

in cui gli $a_{ik}\in Z_3$ cioè possono essere 0, 1, 2.

All'uopo si potrebbero scrivere le equazioni di tutte le coniche con un opportuno programma.

Una conica è degenere se

$$\Delta = \begin{vmatrix} a_{11} & a_{12} & a_{13} \\ a_{12} & a_{22} & a_{23} \\ a_{13} & a_{23} & a_{33} \end{vmatrix} = 0$$

ed è rispettivamente una iperbole, parabola, ellisse a seconda che, calcolando $-\begin{vmatrix} a_{11} & a_{12} \\ a_{12} & a_{22} \end{vmatrix}$ si ha:

$a_{12}^2-a_{11}a_{22}=\square$ quadrato nel campo cioè 1

$a_{12}^2-a_{11}a_{22}=0$

$a_{12}^2-a_{11}a_{22}=\square$ non quadrato nel campo cioè 2

In AG(2,3) un'ellisse ha q+1=4 punti, quelli di una parabola sono q=3, quelli di un'iperbole q-1=2.

Esempi:

La conica $2x^2+y^2+2xy+1=0$, poichè $a_{12}^2-a_{11}a_{22}=1-2=1-2=-1=2$, è un'ellisse e contiene i punti (0,1), (1,1), (2,2), (2,0).

La conica $x^2+y^2+2xy+x=0$, poichè $a_{12}^2-a_{11}a_{22}=0$, è una parabola e contiene i punti (0,0), (2,0), (2,2).

La conica $2x^2+y^2+x=0$, poichè $a_{12}^2-a_{11}a_{22}=-2=1$, è un'iperbole contiene i punti (0,0), (1,0).

La conica $x^2+y^2+2xy+x+y=0$ è semplicemente degenere in quanto (si noti che $1=4$, cioè $\frac{1}{2}=2$)

$$A=\begin{vmatrix}1 & 1 & 2\\ 1 & 1 & 2\\ 2 & 2 & 0\end{vmatrix}=0$$

e si scrive $(x+y)^2+(x+y)=0$ $(x+y)(x+y+1)=0$.

La conica $x^2+y^2+2xy+2x+2y+1=0$ è doppiamente degenere e si scrive $(x+y+2)^2=0$.

Esistono programmi per il calcolo del determinante di una matrice e quindi, mediante essi, si può riconoscere se una conica è degenere e, in caso contrario, se è un'ellisse, una parabola o un'iperbole.

BIBLIOGRAFIA

1. B. SEGRE, *Lectures on modern geometry*, Edizioni Cremonese, Roma 1961.